# $Id$
# Maintainer: Jan Alexander Steffens (heftig) <jan.steffens@gmail.com>
# Contributor: Jan de Groot <jgc@archlinux.org>

pkgname=yelp
pkgver=3.22.0+1+gfabd8eb
pkgrel=1
pkgdesc="Get help with GNOME"
url="https://wiki.gnome.org/Apps/Yelp"
license=(GPL)
arch=(i686 x86_64)
depends=(webkit2gtk yelp-xsl libxslt xz bzip2)
makedepends=(intltool gtk-doc itstool autoconf-archive git xorg-server-xvfb)
groups=(gnome)
_commit=fabd8ebe803e54f654d4930e26dbf695881f52f5  # master
source=("git://git.gnome.org/yelp#commit=$_commit")
sha256sums=('SKIP')

pkgver() {
  cd $pkgname
  git describe --tags | sed 's/-/+/g'
}

prepare() {
  cd $pkgname
  # Fixup missing tag for pkgver()
  git tag -f 3.22.0 9ae6e5a2ad3ea6709d31d110c4cfd160d600ee4f
  NOCONFIGURE=1 ./autogen.sh
}

build() {
  cd $pkgname
  ./configure --prefix=/usr --sysconfdir=/etc --localstatedir=/var \
    --disable-static --enable-gtk-doc
  sed -i -e 's/ -shared / -Wl,-O1,--as-needed\0/g' libtool

  # Documentation generation segfaults without X11
  xvfb-run -a -n 81 -s "-extension GLX -screen 0 1280x1024x24" make
}

package() {
  cd $pkgname
  make DESTDIR="$pkgdir" install
}
